Solving Ordinary Differential Equations Using Maple V

Working with Differential Equations

Many routines for working with ODEs (and PDEs) are contained in the Maple package, DEtools. This should normally be loaded into memory using the 'with' command.

```
> restart:
> with(DEtools);
```

[*AreSimilar*, *Closure*, *DEnormal*, *DEplot*, *DEplot3d*, *DEplot_polygon*, *DFactor*, *DFactorLCLM*, *DFactorsols*, *Dchangevar*, *Desingularize*, *FunctionDecomposition*, *GCRD*, *Gosper*, *Heunsols*, *Homomorphisms*, *IVPsol*, *IsHyperexponential*, *LCLM*, *MeijerGsols*, *MultiplicativeDecomposition*, *ODEInvariants*, *PDEchangecoords*, *PolynomialNormalForm*, *RationalCanonicalForm*, *ReduceHyperexp*, *RiemannPsols*, *Xchange*, *Xcommutator*, *Xgauge*, *Zeilberger*, *abelsol*, *adjoint*, *autonomous*, *bernoullisol*, *buildsol*, *buildsym*, *canoni*, *caseplot*, *casesplit*, *checkrank*, *chinisol*, *clairautsol*, *constcoeffsols*, *convertAlg*, *convertsys*, *dalembertsol*, *dcoeffs*, *de2diffop*, *dfieldplot*, *diff_table*, *diffop2de*, *dperiodic_sols*, *dpolyform*, *dsubs*, *eigenring*, *endomorphism_charpoly*, *equinv*, *eta_k*, *eulersols*, *exactsol*, *expsols*, *exterior_power*, *firint*, *firtest*, *formal_sol*, *gen_exp*, *generate_ic*, *genhomosol*, *gensys*, *hamilton_eqs*, *hypergeomsols*, *hyperode*, *indicialeq*, *infgen*, *initialdata*, *integrate_sols*, *intfactor*, *invariants*, *kovacicsols*, *leftdivision*, *liesol*, *line_int*, *linearsol*, *matrixDE*, *matrix_riccati*, *maxdimsystems*, *moser_reduce*, *muchange*, *mult*, *mutest*, *newton_polygon*, *normalG2*, *ode_int_y*, *ode_y1*, *odeadvisor*, *odepde*, *parametricsol*, *particularsol*, *phaseportrait*, *poincare*, *polysols*, *power_equivalent*, *rational_equivalent*, *ratsols*, *redode*, *reduceOrder*, *reduce_order*, *regular_parts*, *regularsp*, *remove_RootOf*, *riccati_system*, *riccatisol*, *rifread*, *rifsimp*, *rightdivision*, *rtaylor*, *separablesol*, *singularities*, *solve_group*, *super_reduce*, *symgen*, *symmetric_power*, *symmetric_product*, *symtest*, *transinv*, *translate*, *untranslate*, *varparam*, *zoom*] **(1.1)**

Entering Equations

A. First Order ODEs :

```
> diff(y(t),t)+y(t)+x(t)=0;
```

$$\frac{d}{dt} y(t) + y(t) + x(t) = 0 \qquad \textbf{(2.1)}$$

```
> diff(y(t),t)-y(t)=sin(t);
```

$$\frac{d}{dt} y(t) - y(t) = \sin(t) \qquad \textbf{(2.2)}$$

B. Higher Order ODEs :

```
> m*diff(y(t),t$2)+b*diff(y(t),t)+k*y(t)=0;
```

$$m\left(\frac{d^2}{dt^2} y(t)\right) + b\left(\frac{d}{dt} y(t)\right) + k\,y(t) = 0 \qquad \textbf{(2.3)}$$

```
> diff(y(t),t$4)-10*diff(y(t),t$3)+35*diff(y(t),t$2)-50*y(t)+24=5*
  exp(t);
```

$$\frac{d^4}{dt^4} y(t) - 10 \frac{d^3}{dt^3} y(t) + 35 \frac{d^2}{dt^2} y(t) - 50\, y(t) + 24 = 5\, e^{t} \quad \textbf{(2.4)}$$

Solving

A. Single Equations

Example 1 - Response of a Parallel RLC Circuit

```
> R:=10000:
  L:=0.001:
  C:=0.000001:
> diff(V(t),t$2) + (1/(R*C))*diff(V(t),t) + V(t)/(L*C) = 0;
```

$$\frac{d^2}{dt^2} V(t) + 100.0000000 \frac{d}{dt} V(t) + 1.000000000\, 10^9\, V(t) = 0 \quad \textbf{(3.1.1)}$$

A general solution is found using:

```
> dsolve(%, V(t), method=laplace);
```

$$V(t) = \frac{1}{19999950} \left(e^{-50\, t} \left(19999950\, V(0) \cos\left(50 \sqrt{399999}\, t\right) + \sin\left(50 \sqrt{399999}\, t\right) \sqrt{399999}\, \left(D(V)(0) + 50\, V(0)\right) \right) \right) \quad \textbf{(3.1.2)}$$

A particular solution can be found using initial values for V and dV/dt.

```
> dsolve({%%, V(0)=10, D(V)(0)=1}, V(t));
```

$$V(t) = \frac{167 \sqrt{399999}\, e^{-50\, t} \sin\left(50 \sqrt{399999}\, t\right)}{6666650} + 10\, e^{-50\, t} \cos\left(50 \sqrt{399999}\, t\right) \quad \textbf{(3.1.3)}$$

Example 2 - Bessel Equation and Series Solution

```
> x^2*diff(y(t),t$2) + x*diff(y(t),t) + (x^2-1)*y = 0;
```

$$x^2 \left(\frac{d^2}{dt^2} y(t) \right) + x \left(\frac{d}{dt} y(t) \right) + \left(x^2 - 1\right) y = 0 \quad \textbf{(3.2.1)}$$

```
> dsolve({%, y(0)=0, D(y)(0)=1}, y(t), type=series);
```

$$y(t) = t - \frac{1}{2} \frac{1}{x} t^2 - \frac{1}{6} \frac{x^2 - 2}{x^2} t^3 + \frac{1}{24} \frac{2\, x^2 - 3}{x^3} t^4 + \frac{1}{120} \frac{x^4 - 5\, x^2 + 5}{x^4} t^5 + O\left(t^6\right) \quad \textbf{(3.2.2)}$$

Example 3 - Numerical Solution

```
> diff(y(t),t$2) + sin(t)*y(t) = 0;
```

$$\frac{d^2}{dt^2} y(t) + \sin(t)\, y(t) = 0 \quad \textbf{(3.3.1)}$$

```
> soln := dsolve({%, y(0)=2, D(y)(0)=-2}, y(t), type=numeric);
```

$$soln := \textbf{proc}(x\_rkf45)\ \ldots\ \textbf{end proc} \quad \textbf{(3.3.2)}$$

```
> soln(0); soln(Pi/2); soln(Pi); soln(3*Pi/2);
```

$$\left[t = 0.,\ y(t) = 2.,\ \frac{d}{dt} y(t) = -2. \right]$$

$$\left[t = 1.57079632679490,\ y(t) = -1.36964878441444,\ \frac{d}{dt} y(t) = -1.85177381242493 \right]$$

$$\left[t=3.14159265358979,\ y(t)=-2.34660817162069,\ \frac{d}{dt}\,y(t)=0.189671391783838\right]$$

$$\left[t=4.71238898038468,\ y(t)=-3.45326122922634,\ \frac{d}{dt}\,y(t)=-2.47923087286615\right] \quad (3.3.3)$$

B. Systems

Example 4

```
> interface(labelling=false):
  eqn1 := diff(y(t),t) = z(t) - y(t);
  eqn2 := diff(z(t),t) = y(t);
```

$$eqn1 := \frac{d}{dt}\,y(t)=z(t)-y(t)$$

$$eqn2 := \frac{d}{dt}\,z(t)=y(t) \quad (3.4.1)$$

```
> soln := dsolve({eqn1, eqn2, y(0)=0, z(0)=1}, {y(t), z(t)}):
> collect(soln,exp);
```

$$\left\{y(t)=\left(\frac{1}{2}+\frac{\sqrt{5}}{10}\right)\left(\frac{\sqrt{5}}{2}-\frac{1}{2}\right)e^{\frac{(\sqrt{5}-1)t}{2}}+\left(\frac{1}{2}-\frac{\sqrt{5}}{10}\right)\left(-\frac{\sqrt{5}}{2}-\frac{1}{2}\right)e^{-\frac{(\sqrt{5}+1)t}{2}},\ z(t)=\left(\frac{1}{2}+\frac{\sqrt{5}}{10}\right)e^{\frac{(\sqrt{5}-1)t}{2}}+\left(\frac{1}{2}-\frac{\sqrt{5}}{10}\right)e^{-\frac{(\sqrt{5}+1)t}{2}}\right\} \quad (3.4.2)$$

Plotting Differential Equations

A. Single Equations - **Damped Oscillator Model**

```
> eqn5 := m*diff(x(t),t$2) = -b*diff(x(t),t) - k*x(t);
  m:=0.5: b:=1: k:=20:
> DEplot(eqn5, {x(t)}, 0..5, [[x(0)=5, D(x)(0)=-2]],
       title=`Damped Oscillator`);
```

$$eqn5 := m\left(\frac{d^2}{dt^2}\,x(t)\right)=-b\left(\frac{d}{dt}\,x(t)\right)-k\,x(t)$$



B. Systems of Equations - Dynamical System Model

```
> eqn5 := diff(y(t),t)=-y(t)-x(t);
```

$$eqn5 := \frac{d}{dt}\, y(t) = -y(t) - x(t) \tag{4.1}$$

```
> eqn6 := diff(x(t),t)=y(t);
```

$$eqn6 := \frac{d}{dt}\, x(t) = y(t) \tag{4.2}$$

```
> ic1 := x(0)=0, y(0)=5;
> ic2 := x(0)=0, y(0)=-5;
```

$$ic1 := x(0) = 0, y(0) = 5$$

$$ic2 := x(0) = 0, y(0) = -5 \tag{4.3}$$

```
> DEplot({eqn5,eqn6}, [x(t),y(t)], -5..5, [[ic1],[ic2]], title=
  `Spiral`);
```



Phase Portraits

```
> eqn3 := diff(y(x),x)=-y(x)-x^2;
```

$$eqn3 := \frac{\mathrm{d}}{\mathrm{d}x}\, y(x) = -y(x) - x^2 \qquad \textbf{(5.1)}$$

```
> phaseportrait(eqn3, y(x), x=-1..2.5,
      [[y(0)=0],[y(0)=1],[y(0)=-1]], title=`Asymptotic Solution`,
      color=magenta, linecolor=[red,blue,green]);
```


Asymptotic Solution
y(x)
x
3
2
1
0
−1
−2
−3
−1
1
2